
6. GCD(최대 공약수) 계산
1. 소스
	소스 코드

	include irvine32.inc

CalcGCD     PROTO   x:DWORD, y:DWORD        ; CalcGCD 프로토 타입

ABS         PROTO   x:DWORD                 ; ABS 프로토 타입

.data

input_msg   byte    "Input num ==> ", 0     ; 입력 메시지

res_msg1    byte    " GCD ", 0              ; 결과 출력 메시지 1

res_msg2    byte    " = ", 0                ; 결과 출력 메시지 2

val         sdword  ?, ?                    ; 사용자가 입력한 값 배열

.code

main    PROC

    mov     ecx, 2                  ; 정수 두개를 입력 받음

    mov     esi, 0                  ; esi는 val의 오프셋. 0으로 초기화

L:

    mov     edx, offset input_msg

    call    WriteString             ; 입력 요청 메시지 출력

    call    ReadInt                 ; 사용자에게 정수 입력

    mov     val[esi], eax           ; val에 저장

    add     esi, 4                  ; esi를 다음 오프셋으로 증가

    loop    L                       ; 푸프 반복

    mov     eax, val[0]

    call    WriteInt                ; val1 출력

    mov     edx, offset res_msg1

    call    WriteString             ; GCD 출력

    mov     eax, val[4]

    call    WriteInt                ; val2 출력

    mov     edx, offset res_msg2

    call    WriteString             ; = 출력

    INVOKE  CalcGCD, val[0], val[4] ; GCD 계산

    call    WriteInt                ; 계산 결과 출력

    exit                            ; 프로그램 종료

main    ENDP

ABS     PROC,                       ; ABS 함수

            x:DWORD 

    mov     eax, x                  ; eax를 x로 초기화

    cmp     x, 0                    ; x를 0과 비교

    jge     Positive                ; 0이상인 경우 Positive로 점프

    neg     eax                     ; 0보다 작은 경우 2의 보수로 반전

Positive:

    ret

ABS     ENDP

CalcGCD PROC,                       ; GCD 계산 함수

            x:DWORD, 

            y:DWORD

    LOCAL   n:DWORD                 ; 지역 변수

    INVOKE  ABS, x

    mov     x, eax                  ; x = ABS(x)

    INVOKE  ABS, y

    mov     y, eax                  ; y = ABS(y)

L:

    mov     edx, 0

    mov     eax, x

    mov     ebx, y

    div     ebx                     ; x / y

    mov     eax, y

    mov     x, eax                  ; y = x

    mov     y, edx                  ; y = x % y

    cmp     edx, 0                  ; 나머지가 0인지 비교

    ja      L                       ; 0보다 크면 루프 반복

    mov     eax, x                  ; eax에 x 대입

    ret

CalcGCD ENDP

END main


2. 실행 결과
아래 화면은 (2,3), (5,90), (-30,5), (90,11)의 최대 공약수를 구하는 프로그램 실행 결과를 보여준다.

그림 1 최대 공약수 실행 결과
